

# Работа с математическими функциями

# 11

Длинный список математических функций, которые предоставляет Excel, сразу же позволяет оценить глубину возможностей этой программы. Существуют функции, выполняющие простые математические операции, например вычисление абсолютного значения числа, поиск наибольшего общего делителя или наименьшего общего кратного, извлечение квадратного корня или суммирование. Кроме того, существует большое количество сложных функций для выполнения любых операций, например умножение матриц, работа с многочленами и задачи оптимизации методом наименьших квадратов. Не все математические функции Excel полезны в контексте бизнес-моделей, но все же таких функций на удивление много. Например, такие операции, как округление чисел и генерация случайных чисел имеют свое применение для решения деловых задач.

В таблице 11.1 перечислены математические функции Excel, но это не исчерпывающий список. Вместо этого сосредоточимся на тех функциях, которые будут полезны в бизнес-моделях. Кроме того, не забывайте, что в Excel есть много статистических функций, которые рассматриваются в главе 12.

## В ЭТОЙ ГЛАВЕ

Особенности функций округления Excel.....	288
Суммирование значений .....	295
Функция ОСТАТ() .....	297
Генерация случайных чисел.....	301

Таблица 11.1. Математические функции Excel

Функция	Описание
ABS (число)	Возвращает абсолютное значение (модуль) заданного числа
EXP (число)	Возвращает число $e$ в степени <i>число</i>
LN (число)	Возвращает натуральный логарифм заданного числа
LOG (число; [основание])	Возвращает логарифм заданного числа по заданному основанию
LOG10 (число)	Возвращает десятичный логарифм числа
ЗНАК (число)	Возвращает сигнум, или знак, заданного числа (1 — положительное число, 0 — ноль, -1 — отрицательное число)
КОРЕНЬ (число)	Возвращает арифметический квадратный корень заданного числа
КОРЕНЬПИ (число)	Возвращает арифметический квадратный корень из заданного числа, умноженного на число $\pi$ . Данная функция отображается в категории <b>Мат.</b> и <b>тригонометрия</b> при подключении надстройки <b>Пакет анализа</b> ; а при установке надстройки <b>Analysis Toolpak — VBA</b> в категории <b>Математические</b> отображается аналогичная функция <b>VBA SqrtPI (число)</b>
МОБР (массив)	Возвращает обратную матрицу для матрицы, заданной как <i>массив</i>
МОПРЕД (массив)	Возвращает определитель матрицы, заданной как <i>массив</i>
МУЛЬТИНОМ (число1; ; [число2; ...])	Возвращает мультиномиальный коэффициент множества чисел (мультиномиальный коэффициент вычисляется как отношение факториала суммы заданных чисел к произведению факториалов всех заданных чисел). Эта функция отображается в категории <b>Мат.</b> и <b>тригонометрия</b> при подключении надстройки <b>Пакет анализа</b> ; а при установке надстройки <b>Analysis Toolpak — VBA</b> в категории <b>Математические</b> отображается аналогичная функция <b>VBA Multinomial (число1; [число2; ...])</b>
МУМНОЖ (массив1; массив2)	Возвращает произведение матриц, заданных как <i>массив1</i> и <i>массив2</i>
НЕЧЕТ (число)	Округляет <i>число</i> до ближайшего нечетного числа
НОД (число1; [число2; ...])	Возвращает наибольший общий делитель заданных чисел. Эта функция отображается в категории <b>Мат.</b> и <b>тригонометрия</b> при подключении надстройки <b>Пакет анализа</b> ; а при установке надстройки <b>Analysis Toolpak — VBA</b> в категории <b>Математические</b> отображается аналогичная функция <b>VBA Gcd (число1; [число2; ...])</b>

Продолжение табл. 11.1

Функция	Описание
НОК( <i>число1</i> ; [ <i>число2</i> ; ...])	Возвращает наименьшее общее кратное заданных чисел. Эта функция отображается в категории <b>Мат.</b> и <b>тригонометрия</b> при подключении надстройки <b>Пакет анализа</b> ; а при установке надстройки <b>Analysis Toolpak — VBA</b> в категории <b>Математические</b> отображается аналогичная функция <b>VBA Lcm(<i>число1</i>; [<i>число2</i>; ...])</b>
ОКРВВЕРХ( <i>число</i> ; <i>точность</i> )	Округляет заданное <i>число</i> вверх до ближайшего числа, кратного заданной <i>точности</i>
ОКРВНИЗ( <i>число</i> ; <i>точность</i> )	Округляет заданное <i>число</i> до ближайшего числа, кратного заданной <i>точности</i>
ОКРУГЛ( <i>число</i> ; <i>число_разрядов</i> )	Округляет заданное число до указанного числа разрядов
ОКРУГЛВВЕРХ( <i>число</i> ; <i>число_разрядов</i> )	Округляет заданное <i>число</i> до ближайшего большего по модулю, до указанного числа разрядов
ОКРУГЛВНИЗ( <i>число</i> ; <i>число_разрядов</i> )	Округляет заданное <i>число</i> до ближайшего меньшего по модулю, до указанного числа разрядов
ОКРУГЛТ( <i>число</i> ; <i>множитель</i> )	Округляет <i>число</i> с желаемой точностью: до числа, кратного заданному <i>множителю</i> . Эта функция отображается в категории <b>Мат.</b> и <b>тригонометрия</b> при подключении надстройки <b>Пакет анализа</b> ; а при установке надстройки <b>Analysis Toolpak — VBA</b> в категории <b>Математические</b> отображается аналогичная функция <b>VBA MRound(<i>число</i>; <i>множитель</i>)</b>
ОСТАТ( <i>число</i> ; <i>делитель</i> )	Возвращает остаток от деления <i>числа</i> на <i>делитель</i>
ОТБР( <i>число</i> ; [ <i>число_разрядов</i> ])	Усекает число, оставляя указанное число разрядов после десятичного разделителя
ПИ()	Возвращает число $\pi$
ПРОИЗВЕД( <i>число1</i> ; [ <i>число2</i> ; ...])	Перемножает заданные числа
ПРОМЕЖУТОЧНЫЕ.ИТОГИ( <i>номер_функции</i> ; <i>ссылка1</i> ; [ <i>ссылка2</i> , ...])	Возвращает промежуточные итоги в список или базу данных
РИМСКОЕ( <i>число</i> ; [ <i>форма</i> ])	Преобразует заданное арабское <i>число</i> в эквивалентное ему римское число (в текстовом формате)

Продолжение табл. 11.1

Функция	Описание
РЯДСУММ ( <i>x</i> ; <i>m</i> ; <i>n</i> ; <i>коэффициенты</i> )	Возвращает сумму степенного ряда, вычисленную по формуле. Эта функция отображается в категории <b>Мат. и тригонометрия</b> при подключении надстройки <b>Пакет анализа</b> ; а при установке надстройки <b>Analysis Toolpak — VBA</b> в категории <b>Математические</b> отображается аналогичная функция VBA <code>SeriesSum(X; m; n; коэффициенты)</code>
СЛУЧМЕЖДУ ( <i>нижн_гран</i> ; <i>верхн_гран</i> )	Возвращает случайное число в пределах от заданной нижней границы до заданной верхней границы. Эта функция отображается в категории <b>Мат. и тригонометрия</b> при подключении надстройки <b>Пакет анализа</b> ; а при установке надстройки <b>Analysis Toolpak — VBA</b> в категории <b>Математические</b> отображается аналогичная функция VBA <code>Randbetween(нижн_гран; верхн_гран)</code>
СЛУЧИС ()	Возвращает случайное число от 0 до 1
СТЕПЕНЬ ( <i>число</i> ; <i>степень</i> )	Возводит <i>число</i> в указанную <i>степень</i>
СУММ( <i>число1</i> ; [ <i>число2</i> ; ...])	Суммирует аргументы
СУММЕСЛИ ( <i>диапазон</i> ; <i>критерий</i> ; <i>диапазон_суммирования</i> )	Суммирует только те значения в указанном <i>диапазоне</i> , которые соответствуют заданному <i>критерию</i>
СУММПРОИЗВ ( <i>массив1</i> ; <i>массив2</i> ; [ <i>массив3</i> ; ...])	Перемножает соответствующие элементы заданных массивов и суммирует результаты
СУММКВ ( <i>число1</i> ; [ <i>число2</i> ; ...])	Возвращает сумму квадратов аргументов
СУММКВРАЗН ( <i>массив_x</i> ; <i>массив_y</i> )	Возвращает сумму квадратов разностей соответствующих значений в заданных массивах
СУММРАЗНКВ ( <i>массив_x</i> ; <i>массив_y</i> )	Возвращает сумму разностей квадратов соответствующих значений в заданных массивах
СУММСУММКВ ( <i>массив_x</i> ; <i>массив_y</i> )	Возвращает сумму сумм квадратов соответствующих значений в заданных массивах
ФАКТР ( <i>число</i> )	Возвращает факториал заданного числа
ЦЕЛОЕ ( <i>число</i> )	Округляет <i>число</i> до ближайшего меньшего целого числа
ЧЕТН ( <i>число</i> )	Округляет число до ближайшего четного числа

Окончание табл. 11.1

Функция	Описание
ЧАСТНОЕ (числитель; знаменатель)	QUOTIENT (числитель; знаменатель)
ЧИСЛОКОМБ (число; число_выбранных)	Возвращает количество сочетаний — число возможных способов выбрать заданное число_выбранных объектов из указанного числа объектов

Хотя тригонометрические функции Excel в этой книге не рассматриваются, в таблице 11.2 перечислены все функции из этой категории. При работе с этими функциями следует помнить ряд основных моментов.

- В синтаксисе всех функций *число* — это угол, выраженный в радианах.
- Если исходное значение угла выражено в градусах, его можно преобразовать в радианы, умножив на  $\text{ПИ}()$  и разделив на 180. Кроме того, можно использовать функцию РАДИАНЫ (*УГОЛ*), которая преобразует *УГОЛ* из градусов в радианы.
- Тригонометрические функции возвращают значение в радианах. Если результат необходимо преобразовать в градусы, умножьте его на 180 и разделите на  $\text{ПИ}()$ . Кроме того, можно использовать функцию ГРАДУСЫ (*УГОЛ*), которая преобразует *УГОЛ* из радианов в градусы.

Таблица 11.2. Тригонометрические функции Excel

Функция	Описание
ACOS ( <i>число</i> )	Возвращает значение в радианах между 0 и $\pi$ , которое равняется арккосинусу заданного числа ( <i>число</i> должно быть в диапазоне от -1 до 1)
ACOSH ( <i>число</i> )	Возвращает значение в радианах, которое равняется гиперболическому арккосинусу заданного числа ( <i>число</i> должно быть больше или равно 1)
ASIN ( <i>число</i> )	Возвращает значение в радианах от $-\pi/2$ до $\pi/2$ , которое равняется арксинусу заданного числа ( <i>число</i> должно быть в диапазоне от -1 до 1)
ASINH ( <i>число</i> )	Возвращает значение в радианах от $-\pi/2$ до $\pi/2$ , которое равняется гиперболическому арксинусу заданного числа
ATAN ( <i>число</i> )	Возвращает значение в радианах от $-\pi/2$ до $\pi/2$ , которое равняется арктангенсу заданного числа
ATAN2 ( <i>x</i> ; <i>y</i> )	Возвращает значение в радианах от $-\pi$ до $\pi$ (не включая границы интервала), которое равняется арктангенсу координат <i>x</i> и <i>y</i>

Окончание табл. 11.2

Функция	Описание
АТАНН ( <i>число</i> )	Возвращает значение в радианах, которое равняется гиперболическому арктангенсу заданного числа ( <i>число</i> должно быть в диапазоне от -1 до 1)
COS ( <i>число</i> )	Возвращает значение в радианах, которое равняется косинусу заданного числа
COSH ( <i>число</i> )	Возвращает значение в радианах, которое равняется гиперболическому косинусу заданного числа
SIN ( <i>число</i> )	Возвращает значение в радианах, которое равняется синусу заданного числа
SINH ( <i>число</i> )	Возвращает значение в радианах, которое равняется гиперболическому синусу заданного числа
TAN ( <i>число</i> )	Возвращает значение в радианах, которое равняется тангенсу заданного числа
TANH ( <i>число</i> )	Возвращает значение в радианах, которое равняется гиперболическому тангенсу заданного числа
ГРАДУСЫ ( <i>угол</i> )	Преобразует <i>УГОЛ</i> из радианов в градусы
РАДИАНЫ ( <i>угол</i> )	Преобразует <i>УГОЛ</i> из градусов в радианы

## Особенности функций округления Excel

Функции округления Excel полезны во многих ситуациях, например при работе с некруглыми ценами, для вычисления расценок за время работы, кратное четверти часа, а также для дополнительной страховки при работе с заведомо целыми числами, такими как численность персонала или количество единиц штучного товара.

Проблема заключается в том, что в Excel есть много разных функций округления, так что сложно разобраться, какая из них больше подходит для текущей задачи. Чтобы вы все же смогли разобраться в работе этих функций, в этом разделе рассматриваются характерные особенности и отличия между 10 функциями округления Excel — ОКРУГЛ(), ОКРУГЛВВЕРХ(), ОКРУГЛВНИЗ(), ОКРУГЛТ(), ОКРВВЕРХ(), ОКРВНИЗ(), ЧЕТН(), НЕЧЕТ(), ЦЕЛОЕ() и ОТБР().

### Функция ОКРУГЛ()

Наиболее популярная функция округления — это функция ОКРУГЛ():

ОКРУГЛ(*число*; *число\_разрядов*)

*число*

Число, которое необходимо округлить

*число\_разрядов*

Целое число, соответствующее количеству разрядов, до которого следует округлить *число*:

<i>число_разрядов</i>	<i>Описание</i>
> 0	Округляет <i>число</i> до указанного количества знаков после десятичного разделителя
0	Округляет <i>число</i> до ближайшего целого числа
< 0	Округляет <i>число</i> до указанного количества разрядов слева от десятичного разделителя

В таблице 11.3 показаны результаты функции ОКРУГЛ() для различных значений аргумента *число\_разрядов*. В данном примере было взято *число*, равное 1234,5678.

**Таблица 11.3. Результаты функции ОКРУГЛ() для различных значений аргумента *число\_разрядов***

<i>число_разрядов</i>	Результат функции ОКРУГЛ(1234,5678; <i>число_разрядов</i> )
3	1234,568
2	1234,57
1	1234,6
0	1235
-1	1230
-2	1200
-3	1000

## Функция ОКРУГЛТ()

Функция ОКРУГЛТ() входит в состав надстройки **Пакет анализа**; в надстройке **Analysis Toolpak — VBA** содержится аналогичная функция **VBA MRound()**.

ОКРУГЛТ(*число*; *множитель*)

*число* Число, которое необходимо округлить

*множитель* Кратность, до которой должно быть округлено *число*

В таблице 11.4 показан ряд примеров применения функции ОКРУГЛТ().

**Таблица 11.4. Примеры функции ОКРУГЛТ()**

<i>число</i>	<i>множитель</i>	Результат функции ОКРУГЛТ()
5	2	6
11	5	10
13	5	15
5	5	5
7,31	0,5	7,5
-11	-5	-10
-11	5	#ЧИСЛО!

## Функции ОКРУГЛВНИЗ() и ОКРУГЛВВЕРХ()

Функции ОКРУГЛВВЕРХ() и ОКРУГЛВНИЗ() очень похожи на функцию ОКРУГЛ(), за исключением того, что они всегда округляют число в одном определенном направлении: функция ОКРУГЛВНИЗ() всегда округляет число по направлению к 0, а функция ОКРУГЛВВЕРХ() всегда округляет число от 0. Эти две функции имеют следующий синтаксис:

ОКРУГЛВВЕРХ(*число*; *число\_разрядов*)  
ОКРУГЛВНИЗ(*число*; *число\_разрядов*)

<i>число</i>	Число, которое необходимо округлить
<i>число_разрядов</i>	Целое число, соответствующее количеству разрядов, до которого следует округлить <i>число</i> :
<i>число_разрядов</i>	<i>Описание</i>
> 0	Округляет <i>число</i> вниз или вверх до указанного количества знаков после десятичного разделителя
0	Округляет <i>число</i> вниз или вверх до ближайшего целого числа
< 0	Округляет <i>число</i> вниз или вверх до указанного количества разрядов слева от десятичного разделителя

В таблице 11.5 показано практическое применение функций ОКРУГЛВНИЗ() и ОКРУГЛВВЕРХ().

**Таблица 11.5. Примеры функций ОКРУГЛВНИЗ() и ОКРУГЛВВЕРХ()**

<i>число</i>	<i>число_разрядов</i>	ОКРУГЛВНИЗ()	ОКРУГЛВВЕРХ()
1,1	0	1	2
1,678	2	1,67	1,68
1234	-2	1200	1300
-1,1	0	-1	-2
-1234	-2	-1200	-1300

## Функции ОКРВВЕРХ() и ОКРВНИЗ()

Функции ОКРВВЕРХ() и ОКРВНИЗ() представляют собой смешанную версию возможностей функций ОКРУГЛТ(), ОКРУГЛВНИЗ() и ОКРУГЛВВЕРХ(). Они имеют следующий синтаксис:

ОКРВВЕРХ(*число*; *точность*)  
ОКРВНИЗ(*число*; *точность*)

<i>число</i>	Число, которое необходимо округлить
<i>точность</i>	Кратность, до которой необходимо выполнить округление



Обе функции округляют *число* до значения, кратного значению *точность*, но выполняют это по-разному.

- `ОКРВВЕРХ()` выполняет округление *по направлению от* нуля. Например, функция `ОКРВВЕРХ(1,56; 0,1)` дает результат 1,6, а функция `ОКРВВЕРХ(-2,33; -0,5)` дает результат -2,5.
- `ОКРВНИЗ()` выполняет округление *по направлению к* нулю. Например, функция `ОКРВНИЗ(1,56; 0,1)` дает результат 1,5, а функция `ОКРВНИЗ(-2,33; -0,5)` возвращает результат -2,0.

#### Внимание

Для функций `ОКРВНИЗ()` и `ОКРВВЕРХ()` оба аргумента должны быть одного знака — в противном случае функции возвращают значение ошибки `#ЧИСЛ!`. Кроме того, при вводе 0 в качестве второго аргумента функции `ОКРВНИЗ()` будет получена ошибка `#ДЕЛ/0!`.

## Определение отчетного квартала, на который приходится дата

При работе с моделями, предназначенными для обработки бюджетных или финансовых показателей, часто необходимо знать, на какой отчетный квартал попадает конкретно взятая дата. Например, формула расчета бюджетных показателей на основе заложенного роста может изменяться в зависимости от квартала.

Функцию `ОКРВВЕРХ()` можно использовать в сочетании с функцией `ДАТЕДИФ()` из главы 10, чтобы вычислить квартал для заданной даты:

```
=ОКРВВЕРХ( (ДАТЕДИФ(НачалоПериода; НужнаяДата; "m") + 1) / 3; 1)
```

➔ О функции `ДАТЕДИФ()` см. раздел "Функция `ДАТЕДИФ()`", с. 270.

В этой формуле *НачалоПериода* — это дата начала отчетного периода (год), а *НужнаяДата* — дата, с которой необходимо работать. В формуле используется функция `ДАТЕДИФ()` с параметром *m*, возвращающая количество месяцев между двумя датами. В формуле к результату прибавляется 1 (чтобы избежать получения ответа "0-й квартал"), после чего полученное значение делится на 3. Применение функции `ОКРВВЕРХ()` к результату дает квартал, на который приходится нужная дата.

## Вычисление даты Пасхи

Как правило, для предприятий редко бывает необходимо вычислять дату, на которую приходится Пасхальное воскресенье, потому что этот день в любом случае является выходным. Тем не менее, следующий за Пасхой понедельник обычно считается выходным днем, и если график работы предприятия зависит от государственных выходных, то во многих ситуациях может быть необходимо получить дату, на которую в том или ином году приходится понедельник после Пасхи.

К сожалению, простого пути вычисления этой даты не существует. Официальная формула для даты Пасхи — это первое воскресенье после первого полнолуния, наступившего после весеннего равноденствия. Математики столетиями пытались вывести формулу для расчета календаря Пасхи, и некоторым из них это удалось (среди них знаменитый математик Карл Фридрих Гаусс). Тем не менее, полученные алгоритмы необычайно сложны.

Вот относительно простая формула, которая работает на основе функции ОКРВНИЗ () и вычисляет дату Пасхи для года из диапазона от 1900 до 2078 (для даты используется формат ММ/ДД/ГГГГ):

```
=ОКРВНИЗ ("5/" & ДЕНЬ(МИНУТА(В1/38)/2 + 56) & "/" & В1; 7)
☞ - 34 + 1
```

Для форматов даты ДД/ММ/ГГГГ применяйте следующую альтернативную формулу:

```
=ОКРВНИЗ(ДЕНЬ(МИНУТА(В1/38)/2 + 56) & "5/" & В1; 7) - 34
```

## Функции ЧЕТН() и НЕЧЕТ()

Функции ЧЕТН () и НЕЧЕТ () округляют значение, полученное в качестве аргумента:

```
ЧЕТН(число)
НЕЧЕТ(число)
```

*число* Число, которое необходимо округлить

Обе функции округляют значение *число* по направлению от нуля с использованием определенных правил.

- Функция ЧЕТН () округляет *число* до ближайшего четного числа. Например, функция ЧЕТН (14; 2) возвращает 16, а функция ЧЕТН (-23) возвращает -24.
- Функция НЕЧЕТ () округляет *число* до ближайшего нечетного числа. Например, функция НЕЧЕТ (58; 1) возвращает результат 59, а функция НЕЧЕТ (-6) возвращает значение -7.

## Функции ЦЕЛОЕ() и ОТБР()

Функции ЦЕЛОЕ () и ОТБР () работают аналогично, так как обе функции можно использовать для преобразования значения в целое число.

```
ЦЕЛОЕ(число)
ОТБР(число; [число_разрядов])
```

*число* Число, которое необходимо округлить

*число\_разрядов* Целое число, соответствующее количеству разрядов, до которого следует округлить *число*:

*число\_разрядов* Описание

> 0 Отбрасывает все знаки после разделителя до указанного числа разрядов

0	Отбрасывает все знаки после десятичного разделителя (принято по умолчанию)
< 0	Преобразует <i>число_разрядов</i> слева от десятичного разделителя в нули

Например, функция ЦЕЛОЕ(6,75) возвращает результат 6, а функция ОТБР(3,6) возвращает результат 3. Однако между этими функциями есть два существенных отличия, на которые следует обратить внимание.

- Для отрицательных значений функция ЦЕЛОЕ() возвращает следующее число по направлению от нуля. Например, функция ЦЕЛОЕ(-3,42) возвращает -4. Если вы просто хотите отбросить дробную часть, используйте вместо этого функцию ОТБР().
- Второй аргумент функции ОТБР(), *число\_разрядов* можно использовать, чтобы определить количество десятичных знаков, которое необходимо оставить после отбрасывания остальных разрядов. Например, функция ОТБР(123,456; 2) возвращает 123,45, а функция ОТБР(123,456; -2) возвращает результат 100.

### Использование округления для предотвращения ошибок в вычислениях

Большинству из нас вполне удобно работать с числами в десятичной системе (действительно, специалистов-компьютерщиков, предпочитающих шестнадцатеричную систему, довольно мало). Тем не менее, компьютеры преимущественно работают в более простой области — двоичной системе исчисления. Поэтому, когда вы вводите значение в ячейку или формулу, Excel преобразует его из десятичного в двоичный эквивалент, выполняет вычисления, а затем преобразует двоичный результат назад в десятичный формат.

Эта процедура отлично работает с целыми числами, поскольку при этом для всех десятичных значений есть точный двоичный эквивалент. Однако у многих нецелых значений нет точного эквивалента в двоичной системе. Excel может работать только с приближением к этим числам, что может привести к ошибкам в формулах. Например, попробуйте ввести в ячейку следующую формулу:

```
=0,01 = (2,02 - 2,01)
```

В этой формуле значение 0,01 сравнивается с выражением 2,02 - 2,01. Конечно, они должны быть равны, но эта формула в Excel дает результат ЛОЖЬ. С чем это связано? Проблема заключается в том, что при преобразовании выражения 2,02 - 2,01 в двоичную систему и назад число получает дополнительную цифру. Чтобы увидеть это, введите в ячейку формулу =2,02 - 2,01 и сформатируйте ее как число с 16 знаками после запятой. Вы должны увидеть такой неожиданный результат:

```
0,0100000000000002
```

Именно лишняя цифра 2 в шестнадцатом знаке после запятой стала причиной неправильного результата показанной выше формулы. Чтобы испра-

вить эту проблему, используйте функцию ОТБР ( ) (или функцию ОКРУГЛ ( ) в зависимости от ситуации), чтобы отбросить лишние цифры справа от десятичного разделителя. Например, показанная ниже формула дает результат ИСТИНА:

=0,01 = ОТБР (2,02 - 2,01; 2)

## Формирование цен

В электронных таблицах можно часто столкнуться с задачей вычисления цены на продукт в прейскуранте, которая рассчитывается на основе производственных затрат и наценки. Если продукт будет продаваться в розницу, то можно формировать цены с определенной дробной частью (к примеру, 95 или 99 копеек — 0,95 или 0,99). Чтобы реализовать такое “округление”, можно использовать функцию ЦЕЛОЕ ( ).

Например, самый простой способ — всегда округлять дробную часть цены до 0,95. Вот формула, выполняющая такую операцию:

=ЦЕЛОЕ (ИсходнаяЦена) + 0,95

Предполагая, что *ИсходнаяЦена* была получена на основе расчетов производственных затрат и необходимой прибыли, формула просто прибавляет 0,95 к целой части числа. (Кроме того, обратите внимание, что если дробная часть исходной цены больше 0,95, то формула все равно будет округлять цену вниз до 0,95).

Еще один вариант — округлять до 0,5 дробную часть цены, если она меньше или равна 0,5, и округлять вверх до 0,95 дробную часть, которая больше 0,5. Вот формула, реализующая такой подход:

=ЗНАЧ (ЦЕЛОЕ (ИсходнаяЦена) & ЕСЛИ (ИсходнаяЦена -  
 <math>\geq 0,5</math>; "0,50"; "0,95"))

В этом примере аналогично берется целая часть от значения *ИсходнаяЦена*. Кроме того, функция ЕСЛИ ( ) выполняет проверку, что дробная часть числа меньше или равна 0,5. В таком случае функция возвращает строку 0,50; в противном случае функция возвращает строку 0,95. Полученный результат присоединяется к целой части числа, и функция ЗНАЧ ( ) выдает числовой результат из этого текста.

### ПРАКТИКУМ

#### Округление времени работ для вычисления стоимости

Идеальное применение функции ОКРУГЛТ ( ) — это округление времени работ, на основе которого рассчитывается стоимость работ, до значения, кратного некоторому числу минут. Например, при выставлении счета время может округляться до ближайшего значения, кратного 15 минутам. Для этого можно использовать функцию ОКРУГЛТ ( ) в следующей общей форме:

ОКРУГЛТ (Время; 0:15)

В данном случае *Время* представляет собой то значение, которое необходимо округлить. Например, следующее выражение дает результат 2 : 15:

```
ОКРУГЛТ (2 : 10 ; 0 : 15)
```

Тем не менее, в использовании функции ОКРУГЛТ () для округления значения времени есть один существенный недостаток: большинство предпочитает округлять интервал времени вверх до ближайших 15 минут (или другой базовой единицы для вычислений). Если же компонент минут в аргументе число функции ОКРУГЛТ () окажется меньше, чем половина аргумента множителя, то функция ОКРУГЛТ () округлит значение *вниз* до ближайшего множителя.

Чтобы исправить эту проблему, следует применять вместо функции ОКРУГЛТ () функцию ОКРУГЛВВЕРХ (), поскольку она всегда выполняет округление в сторону от нуля. Вот общее выражение, которое необходимо использовать для округления интервала времени вверх до ближайшего значения, кратного 15 минутам:

```
ОКРУГЛВВЕРХ (Время ; 0 : 15)
```

Здесь *Время* — это снова значение времени, которое необходимо округлить. Например, следующее выражение возвращает значение времени 2 : 15:

```
ОКРУГЛВВЕРХ (2 : 05 ; 0 : 15)
```

## Суммирование значений

Суммирование значений — это одна из самых популярных операций в электронных таблицах, будь то диапазон ячеек, результаты функций, числовые константы или результаты выражений. Excel позволяет складывать значения с использованием оператора сложения (+), но зачастую более удобно получать сумму значений с помощью функции СУММ(), которая рассматривается в этом разделе.

### Функция СУММ()

Функция СУММ() имеет следующий синтаксис:

```
СУММ(число1; [число2; ...])
```

*число1, число2, ...* Значения, которые необходимо сложить

В функцию СУММ() можно ввести до 30 аргументов. Например, следующая формула возвращает результат суммирования значений в трех отдельных диапазонах:

```
=СУММ(A2:A13; C2:C13; E2:E13)
```

### Вычисление накопительных сумм

Во многих электронных таблицах необходимо вычислять накопительные суммы. Например, при расчетах бюджетов нужно отображать накопительные суммы выручки от реализации и издержек за различные периоды отчетного года. Аналогично, при расчете выплат по кредиту вычисляются накопительные суммы процентов и основной суммы долга, которые должны быть выплачены на том или ином этапе пользования кредитом.

Расчет этих накопительных сумм выполняется очень легко. Например, рассмотрим таблицу выплат по кредиту (рис. 11.1). В столбце F вычисляется накопительный процент по кредиту, а в ячейке F6 содержится следующая формула СУММ() :

=СУММ(\$D\$7:D7)

Период	Месяц	Платеж	Процент	Основная часть долга	Общий процент	% выплаты основной суммы
1	январь 04	-234,85	-50,00	-184,85	-50,00	1,85%
2	февраль 04	-234,85	-49,08	-185,77	-99,08	3,71%
3	март 04	-234,85	-48,15	-186,70	-147,23	5,57%
4	апрель 04	-234,85	-47,21	-187,64	-194,44	7,45%
5	май 04	-234,85	-46,28	-188,58	-240,71	9,34%
6	июнь 04	-234,85	-45,33	-189,52	-286,04	11,23%
7	июль 04	-234,85	-44,38	-190,47	-330,43	13,14%
8	август 05	-234,85	-43,43	-191,42	-373,86	15,05%
9	сентябрь 05	-234,85	-42,48	-192,37	-416,24	16,97%

Рис. 11.1. Формулы СУММ() в столбце F вычисляют накопительный процент, который выплачивается по кредиту

#### Примечание

Файл рабочей книги, содержащий примеры из этой главы, можно загрузить с Web-сайта книги.

В этой формуле суммируется единственное значение из ячейки D7, что само по себе не очень полезно. Однако при заполнении с использованием этой формулы диапазона F7 : F54 левая часть диапазона (\$D\$7) в формуле суммы остается без изменений, а правая часть (D7) задана как относительная ссылка и поэтому изменяется. Таким образом, в ячейке F10 будет следующая формула:

=СУММ(\$D\$7:D10)

Если вы хотите разобрать пример до конца, то в столбце G вычисляется процент от общей суммы, который был выплачен на определенный момент. Вот формула, используемая в ячейке G7:

=СУММ(\$E\$7:E7) / \$B\$4 \* -1

Фрагмент формулы СУММ(\$E\$7:E7) вычисляет накопленную выплаченную сумму кредита. Чтобы определить, какой процент от общей основной суммы кредита составляет эта величина, необходимо разделить ее на общую сумму кредита (ячейка B4). Вся формула умножается на -1, чтобы получить положительный процент.

## Суммирование только положительных или только отрицательных значений диапазона

Предположим, вы работаете с диапазоном чисел, в котором есть и положительные, и отрицательные значения, а необходимо вычислить сумму только отрицательных либо только положительных значений. Конечно, можно ввести в функцию СУММ() отдельные ячейки, но существует и более простое решение, основанное на применении массивов.

Чтобы получить сумму отрицательных значений в диапазоне, используйте следующую формулу массива:

```
{=СУММ((диапазон < 0) * диапазон)}
```

В данном примере *диапазон* — это ссылка на диапазон либо именованный диапазон. Условие *диапазон < 0* возвращает ИСТИНА (числовое значение 1) для диапазонов, которые меньше нуля; в противном случае условие возвращает результат ЛОЖЬ (числовое значение 0). Следовательно, в сумму включаются только отрицательные значения.

Аналогично можно использовать следующую формулу массива, чтобы включить в подсчет суммы только положительные значения:

```
{=СУММ((диапазон > 0) * диапазон)}
```

→ Вы можете применить при суммировании значений более сложные критерии, используя функцию СУММЕСЛИ(). См. раздел “Использование функции СУММЕСЛИ()”, с. 376.

## Функция ОСТАТ()

Функция ОСТАТ() вычисляет остаток от деления одного числа на другое. Эта функция более полезна, чем кажется на первый взгляд. Она имеет следующий синтаксис:

```
ОСТАТ(число; делитель)
```

Например, функция ОСТАТ(24; 10) дает результат 4, так как  $24 / 10 = 2$  с остатком 4.

Функция ОСТАТ() хорошо подходит для значений, представляющих собой прогрессию или цикл. Например, дни недели, которые возвращает функция ДЕНЬНЕД(), принимают значения от 1 (воскресенье) до 7 (суббота), а затем цикл начинается снова с 1. Таким образом, следующая формула всегда возвращает целое число, соответствующее дню недели:

```
=ОСТАТ(число; 7) + 1
```

Если *число* — это любое целое число, функция ОСТАТ() возвращает целые значения от 0 до 6, поэтому при прибавлении 1 будут получены числа от 1 до 7.

Аналогичные формулы можно создать для месяцев (от 1 до 12), секунд или минут (от 0 до 59), отчетных кварталов (от 1 до 4) и других задач.

## Усовершенствованная формула для расчета разницы моментов времени

В главе 10 было показано, что при вычитании более раннего времени из более позднего возникают проблемы, если более раннее время приходится раньше полуночи, а более позднее время — после полуночи. Вот выражение, которое было использовано для решения этой проблемы:

```
ЕСЛИ(КонечноеВремя < НачальноеВремя;  
⌘ 1 + КонечноеВремя - НачальноеВремя;  
⌘ КонечноеВремя - НачальноеВремя)
```

➔ Более подробное описание формулы разности двух моментов времени см. в разделе “Вычисление разницы между двумя значениями времени”, с. 278.

Однако значения времени представляют собой циклическую последовательность: это действительные числа от 0 до 1, и после полуночи каждого дня снова начинается отсчет с 0. Следовательно, можно использовать функцию ОСТАТ(), которая значительно упростит формулу для вычисления разности между двумя моментами времени:

```
=ОСТАТ(КонечноеВремя - НачальноеВремя; 1)
```

Эта формула будет работать для любых значений *КонечноеВремя* и *НачальноеВремя*, если момент *КонечноеВремя* — позже, чем *НачальноеВремя*.

## Суммирование значений в каждой *n*-й строке

В зависимости от структуры таблицы может возникнуть необходимость подсчитывать сумму значений в каждой *n*-й строке, где *n* — некоторое целое число. Например, чтобы получить сумму по выборке, можно включать в сумму только каждую 5-ю или 10-ю ячейку общей совокупности.

Для этого можно применить функцию ОСТАТ() к результату функции СТРОКА(), как в этой формуле массива:

```
{=СУММ(ЕСЛИ(ОСТАТ(СТРОКА(Диапазон); n) = 1; Диапазон; 0))}
```

Для каждой *n*-й ячейки в указанном диапазоне функция ОСТАТ(СТРОКА(Диапазон); *n*) возвращает 1. В этом случае к сумме прибавляется значение соответствующей ячейки; в противном случае прибавляется 0. Другими словами, формула подсчитывает сумму значений в первой строке диапазона, затем прибавляет к этому результату сумму значений в (*n*+1)-й строке и т.д. Если вместо этого необходимо начать со 2-й строки диапазона и прибавлять далее (*n*+2)-ю строку и так далее, то следует сравнивать функцию ОСТАТ() с 2, как показано ниже:

```
{=СУММ(ЕСЛИ(ОСТАТ(СТРОКА(Диапазон); n) = 2; Диапазон; 0))}
```



**Первый частный случай: суммирование только по нечетным строкам**

Если необходимо подсчитать сумму только по нечетным строкам в таблице, используйте следующий вариант формулы:

$$\{=\text{СУММ}(\text{ЕСЛИ}(\text{ОСТАТ}(\text{СТРОКА}(\text{Диапазон}); 2) = 1; \text{Диапазон}; 0))\}$$
**Второй частный случай: суммирование только по четным строкам**

Чтобы подсчитать сумму только по четным строкам, необходимо просуммировать ячейки, в которых функция  $\text{ОСТАТ}(\text{СТРОКА}(\text{Диапазон}); 2)$  возвращает 0:

$$\{=\text{СУММ}(\text{ЕСЛИ}(\text{ОСТАТ}(\text{СТРОКА}(\text{Диапазон}); 2) = 0; \text{Диапазон}; 0))\}$$
**Определение високосного года**

Если необходимо определить, является ли тот или иной год високосным, можно использовать для этого функцию  $\text{ОСТАТ}()$ . Високосные года — это, за редким исключением, года, которые делятся на 4 без остатка. Поэтому год (обычно) является високосным, если следующая формула возвращает 0:

$$=\text{ОСТАТ}(\text{год}; 4)$$

В данном случае *год* должен быть представлен в виде четырехзначного числа. Эта формула работает для значений с 1901 по 2099 — этого диапазона значений в большинстве случаев должно быть достаточно. Формула не работает для годов 1900 и 2100, так как эти года, хотя они и делятся на 4 без остатка, не являются високосными. Общее правило для определения високосного года формулируется так: год является високосным, если он делится на 4 и при этом не делится на 100, за исключением случаев, когда год делится на 400. Следовательно, так как года 1900 и 2100 делятся на 100, но не делятся на 400, эти два года не високосные. Однако 2000-й год является високосным. Если вы хотите использовать общую формулу для определения високосного года, она выглядит следующим образом:

$$=(\text{ОСТАТ}(\text{год}; 4) = 0) - (\text{ОСТАТ}(\text{год}; 100) = 0) \\ \& + \text{ОСТАТ}(\text{год}; 400) = 0)$$

Три компонента этой формулы, которые выполняют сравнение результата функции  $\text{ОСТАТ}()$  с 0, возвращают значение 1 или 0. Следовательно, результат этой формулы всегда будет равен 0 для високосных годов и не равен нулю для всех остальных годов.

**Создание попеременного затенения строк**

*Попеременное затенение строк* — это такое форматирование таблицы, при котором строки таблицы поочередно затеняются светлым и более темным цветом (например, светло-серым и темно-серым цветом). Этот тип затенения строк таблицы часто используется в бухгалтерских бланках и других типах финансовой документации; он может быть полезен вообще для любой таблицы, где данные представлены по строкам, так как позволяет более легко отличать строку от соседних строк. Пример такой таблицы показан на рис. 11.2.

№	Дата	Наименование	Получатель платежа/описание	Категория	Дебет	Кредит	Остаток на счете
1	01.01.2004	Изначальн. внос				5 000,00	5 000,00
2	03.06.2004	Аккумуляторное питание привода со счета	Автомат. перевод		-20,00		5 000,00
3	10.06.2004	"ИП" "Автосервис"	Автомат. перевод		-400,00		5 400,00

Рис. 11.2. В таблице для расчета доходов и расходов используется попеременное затенение строк

Конечно, такой формат нелегко создать вручную.

- Во-первых, чтобы отформатировать большой диапазон вручную, понадобится очень много времени.
- Во-вторых, даже если вы применили формат вручную, при вставке или удалении строки придется переделывать все заново.

Чтобы избежать лишней работы, можно применить хитрый прием, основанный на использовании функции `ОСТАТ()` и условного форматирования Excel. Выполните следующие действия:

1. Выделите область, которую необходимо сформатировать с использованием попеременного затенения.
2. Выберите команду **Формат** ⇒ **Условное форматирование**, чтобы отобразить диалоговое окно **Условное форматирование**.
3. В выпадающем списке **Условие 1** выберите вариант **Формула**.
4. В текстовом поле введите следующую формулу: `=ОСТАТ(СТРОКА()); 2)`
5. Щелкните на кнопке **Формат**, чтобы отобразить диалоговое окно **Формат ячеек**.
6. Перейдите на вкладку **Вид**, выберите цвет, который хотите использовать для темных строк таблицы, и щелкните на кнопке **ОК**, чтобы вернуться к диалоговому окну **Условное форматирование** (рис. 11.3).
7. Щелкните на кнопке **ОК**.

Формула `=ОСТАТ(СТРОКА()); 2)` возвращает значение 1 для строк с нечетными номерами и значение 0 для строк с четными номерами. Так как значение 1 эквивалентно логическому значению **ИСТИНА**, Excel применяет к ячейкам с нечетными номерами условное форматирование и оставляет без изменений строки с четными номерами.



**Рис. 11.3.** Формула `ОСТАТ()` применяет затенение ячейки к каждой второй строке (к первой, третьей, пятой и т.д.)

#### Совет

Если вы хотите создать попеременное затенение для столбцов, в диалоговом окне **Условное форматирование** нужно ввести следующую формулу:

`=ОСТАТ(СТОЛБЕЦ(); 2)`

Если же вы хотите сделать четные строки затененными, а нечетные оставить без затенения, введите в текстовом поле в окне **Условное форматирование** следующую формулу:

`=ОСТАТ(СТРОКА()+1; 2)`

## Генерация случайных чисел

Если вы применяете Excel для имитационного моделирования, то в большинстве задач вы будете пользоваться случайными числами, представляющими собой случайный фактор в задаче. Числа можно и придумать, но тогда существует опасность, что вы бессознательно будете подбирать данные по какому-то принципу. Поэтому числа лучше генерировать случайным образом с использованием функций `СЛЧИС()` и `СЛУЧМЕЖДУ()`.

→ В надстройке **Пакет анализа** также есть встроенная функция генерации случайных чисел; см. раздел "Использование инструмента генерации случайных чисел", с. 338.

### Функция `СЛЧИС()`

Функция `СЛЧИС()` возвращает случайное число, которое больше или равно 0 и меньше 1. Функция `СЛЧИС()` полезна и сама по себе (например, она идеально подходит для генерации случайных значений времени). Тем не менее, в большинстве случаев она используется в различных выражениях для генерации случайных чисел из некоторого диапазона.

В самом простом случае, когда нужно получить случайные числа, большие или равные 0 и меньшие  $n$ , используется следующее выражение:

`СЛЧИС() * n`

Например, данная формула генерирует случайное число в диапазоне от 0 до 30:

`=СЛЧИС() * 30`

Более сложный вариант — когда случайные числа должны быть больше или равны некоторому числу  $m$  и меньше некоторого числа  $n$ . Вот выражение, которое используется в таком случае:

$\text{СЛЧИС}() * (n - m) + m$

Например, следующая формула дает случайное число, большее или равное 100 и меньшее 200:

$=\text{СЛЧИС}() * (200 - 100) + 100$

#### Предупреждение

Функция  $\text{СЛЧИС}()$  не постоянна, в том смысле, что значение, которое она возвращает, меняется при каждом пересчете формул, повторном открытии рабочей книги или редактировании ячеек в таблице. Поэтому чтобы ввести в ячейку неизменное случайное число, введите  $=\text{СЛЧИС}()$ , нажмите клавишу <F9>, чтобы вычислить функцию и получить случайное число, а затем нажмите клавишу <Enter>, чтобы поместить случайное значение в ячейку в числовом представлении.

#### Генерация случайных чисел из $n$ цифр

Во многих случаях необходимо генерировать случайные числа, имеющие определенное количество цифр. Например, нужно создавать для новых клиентов случайный учетный номер из шести цифр или присваивать название временному файлу как случайное восьмизначное число.

Чтобы реализовать такую процедуру, необходимо начать с общей формулы из предыдущего раздела и применить к ней функцию  $\text{ЦЕЛОЕ}()$ , чтобы получить целый результат:

$\text{ЦЕЛОЕ}(\text{СЛЧИС}() * (n - m) + m)$

В данном случае число  $n$  должно быть равно некоторому значению  $10^n$ , а  $m$  — значению  $10^{n-1}$ :

$\text{ЦЕЛОЕ}(\text{СЛЧИС}() * (10^n - 10^{n-1}) + 10^{n-1})$

Например, если необходимо генерировать случайное восьмизначное число, формула имеет следующий вид:

$\text{ЦЕЛОЕ}(\text{СЛЧИС}() * (100000000 - 10000000) + 10000000)$

Это выражение генерирует случайные числа, большие либо равные 10000000 и меньшие либо равные 99999999.

#### Генерация случайных символов

Обычно функция  $\text{СЛЧИС}()$  используется для генерации случайных чисел, но ее можно применить и к генерации текстовых значений. Например, предположим, что необходимо случайным образом генерировать букву алфавита. В русском алфавите 33 буквы, поэтому следует начать с выражения, которое генерирует случайные числа в интервале от 1 до 33:

$\text{ЦЕЛОЕ}(\text{СЛЧИС}() * 33 + 1)$

Далее необходимо подставить это выражение в функцию  $\text{ВЫБОР}()$ , которая выбирает букву в зависимости от полученного числа:

```
=ВЫБОР(ЦЕЛОЕ(СЛЧИС()*33+1); "А"; "Б"; "В"; "Г"; "Д"; "Е";  
"Ё"; "Ж"; "З"; "И"; "Й"; "К"; "Л"; "М"; "Н"; "О"; "П"; "Р";  
"С"; "Т"; "У"; "Ф"; "Х"; "Ц"; "Ч"; "Ш"; "Щ"; "Ъ"; "Ы"; "Ь";  
"Э"; "Ю"; "Я")
```

### Случайная сортировка чисел

Иногда при работе с числами в таблице их необходимо отсортировать случайным образом. Например, если вы хотите выполнить некоторую операцию над случайно выбранным подмножеством данных, случайная сортировка данных в таблице позволяет устранить любой неслучайный фактор, который мог влиять на расположение данных.

Чтобы осуществить случайную сортировку данных, выполните следующие действия:

1. Для простоты предположим, что данные организованы по строкам. Выделите диапазон в первом пустом столбце справа или слева от таблицы. Убедитесь, что выделенный диапазон включает то же число строк, что и таблица.
2. Введите **=СЛЧИС()** и нажмите комбинацию клавиш <Ctrl+Enter>, чтобы ввести формулу для генерации случайного числа во все ячейки выделенного диапазона.
3. Выберите команду **Сервис**⇒**Параметры**. Перейдите на вкладку **Вычисления**, установите флажок для параметра **Вручную** и щелкните на кнопке **ОК**.
4. Выделите диапазон, который включает данные и столбец с полученными случайными значениями.
5. Выберите команду **Данные**⇒**Сортировка**, чтобы отобразить диалоговое окно **Сортировка диапазона**.
6. В списке **Сортировать по** выберите столбец, который содержит случайные значения.
7. Щелкните на кнопке **ОК**.

Эта процедура указывает Excel на необходимость сортировки выделенного диапазона согласно случайным значениям, то есть сортировка будет выполняться случайным образом. На рис. 11.4 показан пример. Значения данных находятся в столбце А, значения **СЛЧИС()** находятся в столбце В, а диапазон А2:В21 был отсортирован по столбцу В.

Значения	Значения СЛУЧМЕЖДУ
2	0,166666667
3	0,666666667
4	0,3748967
5	0,600000000
6	0,220000000
7	0,600000000
8	0,704518889
9	0,748625101
10	0,036883999
11	0,596877249
12	0,740999903
13	0,801701005
14	0,102389088
15	0,301603071
16	0,000000000
17	0,309614903
18	0,854347077
19	0,999999999

**Рис. 11.4.** Чтобы случайным образом отсортировать значения данных, добавьте к таблице столбец формул СЛУЧМЕЖДУ() и отсортируйте весь диапазон по случайным значениям

## Функция СЛУЧМЕЖДУ()

Если вы подключили надстройку **Пакет анализа**, то становится доступной функция **СЛУЧМЕЖДУ()**, которая дает возможность упростить работу с наборами случайных чисел. Функция **СЛУЧМЕЖДУ()** позволяет определить нижнюю и верхнюю границы диапазона и возвращает случайное число, попадающее в пределы этого диапазона:

**СЛУЧМЕЖДУ(нижн\_гран; верхн\_гран)**

**нижн\_гран** Минимально возможное случайное число. (Excel генерирует случайное число, которое будет больше или равно данному числу)

**верхн\_гран** Максимально возможное случайное число. (Excel генерирует случайное число, которое будет меньше или равно данному числу)

Например, следующая формула возвращает случайное число от 0 до 59:  
 =СЛУЧМЕЖДУ(1; 59)

## Дополнительно

- Excel предоставляет большой выбор статистических функций для вычисления среднего значения, максимального и минимального значений, стандартных отклонений и других характеристик ряда данных. См. главу 12, “Работа со статистическими функциями”, с. 307.
- Чтобы изучить создание сложных распределений случайных чисел, см. раздел “Использование инструмента генерации случайных чисел”, с. 338.

- Excel позволяет автоматически подсчитать промежуточные суммы по списку данных. Этот прием см. в разделе “Создание автоматических промежуточных итогов”, с. 369.
- Функция СУММЕСЛИ() позволяет применять сложные критерии для выполнения суммирования. См. раздел “Использование функции СУММЕСЛИ()”, с. 376.